
Abstract 
 
This paper will show the advanced characteristics of the Cray X1 and discuss how they are used to 

achieve high-utilized GFLOPS on many real world applications. On most MPP systems, other than the 
Earth Simulator and the Cray Inc. X1, advanced scientific applications do not obtain a high percentage of 
peak performance, in some cases less than 2%. When this small percentage of peak is attained on the 
processor, one needs to have more individual processors to achieve a TFLOP of sustained performance. 
Larger numbers of processors result in a tremendous burden on the interconnect. Here again MPPs other 
than the Earth Simulator and the X1 do not have the interconnect to support the increased number of 
processors. Combining low processor performance with insufficient scaling results in less than desired 
performance for many applications. 

 



 
Introduction 
 
The Cray X1 is a multiple processor system with a high speed interconnect and vector processing 

nodes. In addition to these traditional architectural features it has the ability to globally access remote 
memory as the Cray T3D and T3E. This paper will demonstrate the power of this system by showing 
restructurings that enable applications to take full advantage of the Cray X1.  

 
The applications used to illustrate the requirement optimizations are 1) A widely used Ocean 

Modeling Program developed by LANL under the CHAMMP program – POP [1] and 2) An Adaptive 
Mesh Refinement Code (AMR) developed by Science Applications Inc – SAGE. Each code required 
several optimizations to achieve the desired result and in both cases the percent of Peak Performance on 
the X1 was 30-60 percent. 

 
The communication between the vector nodes can be MPI [2], SHMEM [3] or Co-Array Fortran 

[4,5]. All three of these messaging paradigms can be mixed in the same application. In the POP code, Co-
Arrays were added to a SHMEM version of POP and in the SAGE code, Co-Arrays were added to a MPI 
code. The differences of these messaging systems will be discussed later 

 
Cray X1 
 
The Cray X1 should be viewed as a MPP with vector nodes with a slight twist. The default compute 

unit of the X1 is a MSP (Multi-streaming processor). A MSP is comprised of four SSPs (Single Streaming 
processor). When working in MSP mode the compiler uses the four SSPs for low-level parallelism. The 
four SSPs have extremely low synchronization costs and are either used for dividing up a single vector 
DO loop or the SSPs can be used on an outer DO loop with vectorization on an inner DO loop. The 
advantage of this should be obvious. Shorter loops should be vectorized for a single SSP and another loop 
parallelized over 4 or longer loops can be divided among the four SSPs.  

 
There is a third alternative for using SSPs. Messaging can be performed between SSPs rather than 

MSPs. If an application cannot use the MSP effectively it can use the SSP as the processor. If an 
application does not scale to higher numbers of processors, it can use the MSP as the processor.  

 
The X1’s SMP node contains 4 MSP or 16 SSPs. OpenMP can be used on the node and it works well 

with any of the messaging approaches. Cray Streaming Directives (CSD) is a directive that can be used to 
control SSPs at a higher level. While these directives are similar to OpenMP directives, they represent a 
different level of parallelism. OpenMP, CSDs and messaging can all coexist in an application. Following 
are potential combinations of use of these directives 

 
1. MPI, SHMEM and Co-Array between the MSPs 
2. MPI, SHMEM and Co-Array between the SSPs 
3. CSD directives between SSPs, MPI, SHMEM and Co-Array between MSPs 
4. OpenMP directives between MSPs, MPI, SHMEM and Co-Array between MSPs 
5. OpenMP directives between SSPs (4-16), MPI, SHMEM and Co-Array between the OpenMP 

units. 
6. CSDs between SSPs (4), OpenMP between MSPs (4), MPI, SHMEM and Co-Array between the 

nodes. 



 
Lots of choices, in the POP code, 1) was used most of the time. Examples will be shown where 3) 

was introduced to optimize a particular nasty bit of code. In SAGE 2) was used throughout the code. As 
on other systems, OpenMP is not the preferred parallelization paradigm on the node, try to stick with 1,2 
or 3. 

 
Few words about vector length, there are 32 – 64 word vector registers in the SSP. Each SSP has two 

vector pipes and one scalar processor. When using the MSP, this translates to 128 –64 word registers and 
8 vector pipes. In SSP mode a vector length of 50-100 is ideal, in MSP mode a vector length of 200-400 is 
preferred. As always, the longer the better. 

 
The X1 has a 2 MB cache that is shared by the 4 SSPs. Fortunately the user has the ability to direct 

the compiler not to cache certain vector arguments, choosing to use the vector registers and saving cache 
for more temporal locality.  

 
This is all we will cover on the X1 architecture. This is really all that is required to understand the 

optimizations discussed later. 
 
POP 
 
POP is a multi-dimensional finite difference code that was originally written for the Connection 

Machine. It uses a significant amount of Fortran 90. This is a tremendous advantage for the X1 because; 
the Cray ftn compiler does an excellent job on optimizing vector array syntax. The model illustrated in 
this paper is the 1-degree ocean model. This gives a grid that is 328 x 240 x 32 levels. POP uses a 
processor grid, where one can align the processors to the grid with 2-D decompositions. The approach 
used is the default MSP and the compiler does an excellent job of parallelizing the outer index across the 
SSPs and vectorizing on the inner index – with a few exceptions. 

 
As in any application, a profile is first obtained to identify those routines that use a bulk of the CPU 

time. This is initially done on one MSP and then later on multiple MSPs. The main culprits are imvmixt, 
impvmixu and hmix_aniso. These three routines are using 30-40% of the time on a single MSP. 
Interestingly these three routines were all developed after the Connection Machine went away. 

 
IMP_VMIX 
 
The two imp_vmix routines are very similar. The inner DO-loop is on the ocean layers and it is 

recursive. The two mesh loops I and J are on the outside. While one would expect the compiler to detect 
the recursion and switch DO loops automatically, the way these routines were written confused the 
compiler. Consider the excerpt in Figure 1. 

 
Numerous reasons exist to confuse a vectorizing compiler. Of course the inner loops on K are 

recursive. The fact that the K loop depends upon the outer mesh loops is definitely a problem. The most 
perplexing are the numerous scalars that carry data from one K loop iteration to the next.  



Figure 1 
 
 
 
 
do j=jphys_b,jphys_e 
do i=iphys_b,iphys_e 
a = afac_t(1)*VDC(i,j,1,mt2) 
d = H1(i,j) + a 
e(1) = a/d 
b = H1(i,j)*e(1) 
if (rhs_present) then 
f(1) = hfac_t(1)*RHS(i,j,n)/d 
else 
f(1) = hfac_t(1)*TRACER(i,j,1,n,newtime)/d 
endif 
f(KMT(i,j)+1:km) = c0 
do k=2,KMT(i,j) 
c = a 
a = afac_t(k)*VDC(i,j,k,mt2) 
if (k == KMT(i,j)) then 
d = hfac_t(k)+b 
else 
d = hfac_t(k)+a+b 
endif 
e(k) = a/d 
b = (hfac_t(k) + b)*e(k) 
if (rhs_present) then 
f(k) = c*f(k-1)/d 
else 
f(k) = (hfac_t(k)*TRACER(i,j,k,n,newtime) + c*f(k-1))/d 
endif 
end do 
do k=KMT(i,j)-1,1,-1 
f(k) = f(k) + e(k)*f(k+1) 
end do 
do k = 1,km 
TRACER(i,j,k,n,newtime) =  
     &      TRACER(i,j,k,n,update_index) + f(k) 

       end do 
       end do ! end of i-j loops 

end do 
 
 
 
 



Our optimization approach is therefore to pull the I loop on the inside and leave the J loop on the 
outside. We will then use CSD directives to instruct the compiler to use the 4 SSPs on the J Do loop and 
vectorize on the I loop.  

 
To switch the I loop with the K loop, we need to run the K loop across its entire range and use an IF 

test to restrict which K iterations are performed. Therefore: 
 
do i=iphys_b,iphys_e 
o 
do k=2,KMT(i,j) 
o 
Gets changed to: 
do k=2,km 
do i=iphys_b,iphys_e 
if(k.le.KMT(I,J)) then 
          
Now we need to deal with the troublesome scalars. These scalars are frequently referred to as wrap 

around scalars. The prescription for change is to propagate them to arrays, in this case the length of the i 
index of the arrays by the k index used in the Do loop. Therefore: 

 
c = a 
a= afac_t(k)*VDC(i,j,k,mt2) 
if (k == KMT(i,j)) then 
d = hfac_t(k)+b 
else 
d = hfac_t(k)+a+b 
endif 
e(k) = a/d 
b = (hfac_t(k) + b)*e(k) 
f(k) = (hfac_t(k)*TRACER(i,j,k,n,newtime) + c*f(k-1))/d 
becomes 
c = a(I,k-1) 
a(I,k) = afac_t(k)*VDC(i,j,k,mt2) 
if (k == KMT(i,j)) then 
d = hfac_t(k)+b(I,k-1) 
else 
d = hfac_t(k)+a(I,k)+b(I,k-1) 
endif 
e(I,k) = a(I,k)/d 
b(I,k) = (hfac_t(k) + b(I,k-1))*e(I,k) 
f(I,k) = (hfac_t(k)*TRACER(i,j,k,n,newtime) + c*f(I,k-1))/d  
 
 
 
Notice that c and d do not need to be promoted and pay attention to the use of k-1 on those elements 

that are being carried from the previous to the present DO K loop. This is the inner loop and we must take 
care of the initialization, combining the restructurings we have the resultant code in Figure 2. 



Figure 2 
 
       do j=jphys_b,jphys_e 

do i=iphys_b,iphys_e 
a(I,1) = afac_t(1)*VDC(i,j,1,mt2) 
d = H1(i,j) + a(I,1) 
e(I,1) = a(I,1)/d 
b(I,1) = H1(i,j)*e(1) 
if (rhs_present) then 
f(I,1) = hfac_t(1)*RHS(i,j,n)/d 
else 
f(I,1) = hfac_t(1)*TRACER(i,j,1,n,newtime)/d 
endif 
end do 
do k=2,km 
do i=iphys_b,iphys_e 
if(k.le.KMT(I,j))then 
c = a(I,k-1) 
a(I,k) = afac_t(k)*VDC(i,j,k,mt2) 
if (k == KMT(i,j)) then 
d = hfac_t(k)+b(I,k-1) 
else 
d = hfac_t(k)+a(I,k)+b(I,k-1) 
endif 
e(I,k) = a(I,k)/d 
b(I,k) = (hfac_t(k) + b(I,k-1))*e(k) 
if (rhs_present) then 
f(I,k) = c*f(I,k-1)/d 
else 
f(I,k) = (hfac_t(k)*TRACER(i,j,k,n,newtime) + c*f(I,k-1))/d 
endif 
else 
f(I,k) = c0 
endif 
end do 
end do 
do k=km,1,-1 
do i=iphys_b,iphys_e 
if(k.le.KMT(I,j)-1) f(I,k) = f(I,k) + e(I,k)*f(I,k+1) 
end do 
end do 
do k = 1,km 
do i=iphys_b,iphys_e 
TRACER(i,j,k,n,newtime) = TRACER(i,j,k,n,update_index) + f(I,k) 
end do 
end do 
end do 



 
 
The resultant code gets excellent vectorization on the I DO loop. The compiler cannot find an outer 

parallel DO loop and it chooses to use the full MSP on the inner I loop. Upon inspection we see that the J 
DO loop could be parallel, once again the complexity within the J DO loop is confusing the compiler. To 
solve this problem we move to Cray Streaming Directives (CSD), similar to OpenMP, but used to override 
compiler’s concerns and force the compiler to parallelize the outer DO loop across the SSPs. When 
placing the CSD directive, you must specify those variables that need to be replicated across each SSP. 
For example, each SSP needs its own (PRIVATE) copy of a – this array holds information from K loop to 
K loop. Each J iteration needs a different copy. The Following Directions placed prior to and after the J 
loop parallelizes on J and now we have extreme efficiency. 

 
!CSD$ PARALLEL DO PRIVATE (a,b,c,d,e,f) 
do j=jphys_b,jphys_e 
 
(Loop Body 
enddo 
!CSD$ END PARALLEL DO 
 
While the do loop indices should also be private, we do not need to mention them, the compiler 

always privatizes the loop indices. 
 
HMIX_ANISO 
 
HMIX_ANISO is a good example of a nice big complex DO loop that has been written just wrong. 

Short inner DO loops, Fortran 90 SELECT, CASE statements. HMIX_ANISO is far too big to include in 
the paper so we will just discuss the technique. 

 
Numerous computations are performed within the I, J DO loops. Unfortunately the code developer 

chooses to place inner loops over quadrants and loop independent CASE statements within the grid loops. 
Rather than inverting the grid loops with the quadrant loops, we choose to split the grid loops at each 
CASE statement and around each quadrant loop. To do this restructuring we must promote numerous 
scalars to arrays across the entire local grid dimension. This restructuring could hurt cache-based systems 
since the array dimensions force memory stores that may not have been performed with the scalars in the 
loop. 

 
In this example we did not require CSD directives since the compiler did an excellent job on 

vectorizing the inner I loops and parallelizing the outer J loops. In all cases the quadrant loop was within 
the J loop to parallelize that code as well. All CASE statements, which where loop independent are now 
outside of all loops. 

 
This was all that was required to vectorize the sequential code on the node. Now we moved to 

running across the MSP using one of the options in the POP code – SHMEM. 
 
We ran the restructured code across 16 MSPs and found that the major time consumer was 

BARRIER. This routine synchronizes the processors between message passing operations. Upon 
inspection most of the BARRIER calls were being performed within the 2-Dimensional conjugate gradient 



solver. On a single MSP this code uses less than 10% of the wall clock time. The rest of the application 
scales extremely well and the 2-dimensional solver that contains 3 global reductions and a nine-point 
stencil do not. Consequently at 16 MSPs the solver uses most of the time. In particular the global sum 
operation of single scalars uses the most of the time. This is where we decide to use Co-Array Fortran. 
The SHMEM code is written as follows: 

 
function global_real_sum (X, MASK) 
real (kind=dbl_kind), dimension(imt,jmt), intent(in) ::  
     &   X,                   ! array to be summed 
     &   MASK                 ! real multiplicative mask 
real (kind=dbl_kind) ::  
&   global_real_sum      ! resulting global sum 
integer (kind=int_kind) ::  
&   i,j                  ! local counters 
reduce_real_local = c0 
do j=jphys_b,jphys_e 
do i=iphys_b,iphys_e 
reduce_real_local = reduce_real_local + X(i,j)*MASK(i,j) 
end do 
end do 
call shmem_barrier_all() 
call shmem_real8_sum_to_all (reduce_real_global, 
     &                             reduce_real_local, 1, 0, 0, nproc_s, 
     &                             work_reduce, isync) 
global_real_sum = reduce_real_global 
end function global_real_sum 
 
SHMEM is a very efficient messaging system; however, on the X1 Co-Arrays are much better 

because the compiler can optimize across the messaging code. Pre-fetching of remote memory references 
can be performed as with local memory fetches. The resultant Co-Array code is using an inefficient 
algorithm for large number of processors summation where the master simply accesses the other 
processors values, adds them together and then passes the result back to the other processors. For larger 
values of processors, a LOG N step algorithm would be used that employs multiple masters and performs 
the operations in parallel. As we scale to larger processors such an approach will be used. Co-Arrays is a 
very simple language. It is just Fortran with a second set of subscripts to denote the number of processors. 
Following is the coding. 

 
 
 
 
 
 
 
 
 
 
 



 
function global_sumx (X, MASK) real (kind=dbl_kind), dimension(imt,jmt), intent(in) :: 
     &   X,MASK                 ! real multiplicative mask 
 
       real (kind=dbl_kind) :: 
     &   global_sumx      ! resulting global sum 
 
       integer (kind=int_kind) :: 
     &   i,j                  ! local counters 
        real (kind=dbl_kind):: 
&  reduce_real_local[NPROC_X,*],reduce_real_global[NPROC_X,*] 
reduce_real_local = c0 
do j=jphys_b,jphys_e 
do i=iphys_b,iphys_e 
reduce_real_local = reduce_real_local + X(i,j)*MASK(i,j) 
end do 
end do 
call sync_images() 
if(my_pe().eq.0)then 
reduce_real_global = c0 
do j = 1,NPROC_Y 
do i = 1,NPROC_X 
reduce_real_global = reduce_real_global +reduce_real_local[i,j] 
enddo 
enddo 
do j = 1,NPROC_Y 
do i = 1,NPROC_X 
reduce_real_global[i,j] = reduce_real_global 
enddo 
enddo 
endif 
call sync_images() 
global_sumx = reduce_real_global 
end function global_sumx 
 
Notice that the Co-Arrays are allocated with a processor grid for the 2-D decomposition. The outer-

most dimension of the processor grid must always be *, only the shape of the grid is required. In this case 
all of the Co-Arrays are scalars that we have replicated on each processor. In the first loop nest we obtain 
the local summations, then we sync the processors and have processor 0 proceed to fetch all local sums, 
sum them together and broadcast them back to the other processors. Notice that I,J go over the processor 
grid not the computation grid. After this operation we sync again and each processor sets its global sum. 

 
The other part of the communication problem in the solver is the nine-point stencil. This code 

performs halo exchanges with North/South and East/West neighbor. Given the two-dimensional processor 
grid and the Co-Array function that gives your neighbors, this turns out to be easy. In the following Co-
Array code, XOUT is a Co-Array that is dimensioned on the entire application grid and with Co-Arrays on 
the processor grid. 



        
 
 
me = this_image(XOUT)  ! Argument must be a Co-Array 
if(me(1).ne.NPROC_X)then 
me1p1 = me(1)+1 
else 
me1p1 = 1 
endif 
if(me(1).ne.1)then 
me1m1 = me(1)-1 
else 
me1m1 = NPROC_X 
endif 
ooo 
call sync_images() 
do n=1,num_ghost_cells 
do j=1,jmt 
XOUT(iphys_e+n,j) = XOUT(iphys_b+n-1,j)[me1p1,me(2)] 
          XOUT(        n,j) = XOUT(iphys_e-num_ghost_cells+n,j) 
     .                                                 [me1m1,me(2)] 
end do 
end do 
call sync_images() 
do n=1,num_ghost_cells 
do i=1,imt 
XOUT(i,jphys_e+n) = XOUT(i,jphys_b+n-1)[me(1),me2p1] 
          XOUT(i,        n) = XOUT(i,jphys_e-num_ghost_cells+n) 
     .                                                 [me(1),me2m1] 
end do 
end do 
call sync_images() 
 
The Co-Array function – this_image() takes a Co-Array as an argument and returns a two-dimension 

processor grid location that gives the processor its coordinates within the processor grid. We then 
implement coding to handle the cyclic boundary conditions. Of course the overhead to determine the 
neighbors only needs to be done once.  

We applied Co-Arrays into an existing SHMEM program by changing two routines. Minimal coding 
changes with dramatic results as show in the following chart comparing X1’s POP performance to other 
popular MPP systems in Figure 3. Notice the scaling of the Baroclinic portion of the code on the X1 
versus the other systems. The Baroclinic is the part of the code that contains the 2-D solver where we used 
Co-Arrays. 





 
 
SAGE 
 
POP was relatively easy compared to the work performed on SAGE. The restructurings of SAGE had 

a very dramatic impact on SAGE performance. SAGE is an Adaptive Mesh Refinement (AMR) code and 
all memory references are through indirect addresses. While these gather/scatters can be very efficient on 
the X1, the data dependency analysis is much more difficult. One task was to assure the compiler, which 
scatters where in fact vectorizable. The approach was to place !dir$ concurrent directives prior to those 
important Do loops that where scattering into memory without repeating indices. In addition to convincing 
the compiler to optimize the scatter loops, the following optimizations were performed: 

 
• Restructuring of the sparse matrix times vector to use the method of jagged diagonals [6,7] 
• Restructuring of major communication with Co-Array Fortran 
• Reordering of indirect scatters that do repeat into strings of non-repeatable indices 
• Numerous miscellaneous optimizations 
 
The restructuring of the sparse matrix times vector was performed in the solver used by SAGE. The 

original code vectorized across the non-zero entries in a particular row of the matrix. The average vector 
length was 6-7 and this was not getting good performance on the X1. The restructuring consisted of 
reordering the matrix to do the multiply on the diagonal of the matrix. The vector length of the diagonal 
was approximately 60,000-80,000 –much better.  

 



When SAGE does inter-processor communication, it packs contiguous arrays, performs 
MPI_SEND/MPI_RECVs and then unpacks the data on the receiving processor. This is necessary for 
building larger messages and hiding latencies. This buffer packing results in increased data motion and 
essentially delivers empty flops. With Co-Array Fortran we were able to ovoid the building of the buffers, 
each processor fetches the data it needs directly from the sending processor and stores the data directly 
into its scatter array. On the X1 these distributed memory gather/scatters bypass cache on both ends and 
uses the vector registers on the receiving end to hold and then store the data on processor. Consider the 
following piece of code: 

 
!       Set Co- array pointers to location of input array, send indices and data location 
!       within input array 
 
        send_scratch%p_send_scratch => input(1:length) 
        rsend_index%p_rsend_index => send_index(1:length) 
        send_low%p_send_low => send_lo(0:maxpe) 
 
!     Sync images 
 
        call sync_images() 
          do n=1,recv_num 
pe = recv_pe(n) 
tc = ilenght(recv_length(pe),pe) 
ll = send_low[pe+1]%p_send_low(mype) 
do l=1,tc 
!dir$ concurrent 
do lll=ilength(l,pe),ilenght(l,pe)-1 
rindex = rsend_index[pe+1]%p_rsend_index(ll) 
output(recv_index(lll))=output(recv_index(lll)) +   & 
send_scratch[pe+1]%p_send_scratch(rindex) 
ll = ll + 1 
enddo ! lll 
enddo ! l 
enddo ! n 
!     Sync images 
 
        call sync_images() 
 
This particular routine is called by every processor with a number of processors to send to and a 

number of processors to receive from. Other arguments passed to the routine contain arrays that hold the 
data to send and other arrays are for receiving data. Since this is an irregular grid all accesses are indirect 
and in this example the remote data is added to the on processor output array.  

 
The Co-Array manual says that a pointer cannot be a Co-Array; however, a pointer can be a member 

of a derived type and the derived type can be a Co-Array.  
 
Notice that each processor directly fetches from the remote array using the remote indices and the 

remote location within the input array. The location, a scalar, the indices, an array of indices are first 



fetched and then the actual data elements are fetched from the remote processor and then the data elements 
are scattered into the local processor’s memory.  The pe+1 is needed in the Co-Array because Co-Array 
count from 1 to number of processors and within sage the processors are numbered from 0 to number of 
processors minus one. 

 
In this example we choose to get the data from the remote node. We could have just as easily put the 

data to the remote processor. Putting the data to many processors is actually faster and would be less data 
dependent; however, after the re-write above, this operation takes less than 1% of the time. 

 
Do loops that gather data from the adjoining cells, both on and off processor, can and typically do 

have indices that repeat. These updates are very large vectors and they end up using a significant amount 
of time. Fortunately the indices are computed once each cycle and used frequently. To enable 
vectorization of these gather/scatter loops an algorithm was designed to identify strings of non-repeating 
indices. The resultant data movement loops therefore became double nested loops where the outer loop 
was over the number of strings and the inner was over the number of elements within the strings. 

 
 
MSP versus SSP 
 
After all of the rewrites to optimize SAGE, experiments were made using MSP and SSP execution. 

On a variety of input cases we found that the MSP runs were typically 2.8- 3.0 times faster than the SSP. 
We also saw that both the MSP and SSP parallel runs were scaling similarly. By using SSP mode rather 
than MSP mode we achieved an increase in speed of 4/3. While running on SSPs is a benefit over MSP on 
the size X1 were are now running, larger systems may require changing back to MSP mode. One would 
expect that the MSP mode, using ¼ the number of processors would scale better on larger systems. 
Changing from SSP to MSP and back is very easy, simply use –Ossp for compiling and loading for SSPs 
and omit this option for MSP mode. 

  
Final tuning of SAGE 
 
As we were squeezing the last of the performance out of SAGE, Barriers started to dominate within 

the solver. An additional performance gain was achieved by introducing a double buffering scheme; so 
that, we could avoid one of the barriers. In the code above notice that we have a sync_images( ) prior to 
and after the data transfer – this is required in this case. In obtaining off processor contributions to the 
matrix in the solver, a temporary array – a Co-Array is used for all processors to put into. Initially this 
array was a single dimensioned array and to keep processors from over-writing this array before the 
receiving processor was finished using it a sync_images( ) was required. By dimensioning the temporary 
array by 2 and alternating the second dimension between 1 and 2 half the barriers in the solver were 
eliminated. 

